<commit_before>dnl -------------------------------------------------------------
dnl nemesis
dnl -------------------------------------------------------------
AC_DEFUN([CONFIGURE_NEMESIS],
[
  AC_ARG_ENABLE(nemesis,
                AS_HELP_STRING([--disable-nemesis],
                               [build without NemesisII API support]),
                [case "${enableval}" in
                  yes|new|v522) enablenemesis=yes ; nemesisversion="v5.22" ;;
                  old|v309) enablenemesis=yes ; nemesisversion="v3.09" ;;
                  no) enablenemesis=no  ; nemesisversion=no ;;
                  *) AC_MSG_ERROR(bad value ${enableval} for --enable-nemesis) ;;
                esac],
                [enablenemesis=$enableexodus ; # if unspecified, depend on exodus 
                 if (test "x$exodusversion" = "xv5.22"); then
                   nemesisversion="v5.22"
                 else
                   nemesisversion="v3.09"
                 fi])

  # Trump --enable-nemesis with --disable-mpi
  if (test "x$enablempi" = xno); then
    nemesisversion=no
  fi

  case "${nemesisversion}" in

      "v3.09")
        # The NEMESIS API is distributed with libmesh, so we don't have to guess
        # where it might be installed...
        NEMESIS_INCLUDE="-I\$(top_srcdir)/contrib/nemesis/$nemesisversion"
        AC_DEFINE(HAVE_NEMESIS_API, 1, [Flag indicating whether the library will be compiled with Nemesis support])
        AC_MSG_RESULT(<<< Configuring library with Nemesis version $nemesisversion support >>>)
        ;;

      "v5.22")
        NEMESIS_INCLUDE="-I\$(top_srcdir)/contrib/nemesis/$nemesisversion/nemesis"
        AC_DEFINE(HAVE_NEMESIS_API, 1, [Flag indicating whether the library will be compiled with Nemesis support])
        AC_MSG_RESULT(<<< Configuring library with Nemesis version $nemesisversion support >>>)
        ;;

      *)
        NEMESIS_INCLUDE=""
        enablenemesis=no
        ;;
  esac

  AC_CONFIG_FILES([contrib/nemesis/v3.09/Makefile])
  AC_CONFIG_FILES([contrib/nemesis/v5.22/nemesis/Makefile])<commit_msg>Use AS_CASE and AS_IF in nemesis.m4<commit_after>dnl -------------------------------------------------------------
dnl nemesis
dnl -------------------------------------------------------------
AC_DEFUN([CONFIGURE_NEMESIS],
[
  AC_ARG_ENABLE(nemesis,
                AS_HELP_STRING([--disable-nemesis],
                               [build without NemesisII API support]),
                [AS_CASE("${enableval}",
                         [yes|new|v522], [enablenemesis=yes ; nemesisversion="v5.22"],
                         [old|v309],     [enablenemesis=yes ; nemesisversion="v3.09"],
                         [no],           [enablenemesis=no  ; nemesisversion=no],
                         [AC_MSG_ERROR(bad value ${enableval} for --enable-nemesis)])],
                [
                  # if unspecified, depend on exodus
                  enablenemesis=$enableexodus ;
                  AS_IF([test "x$exodusversion" = "xv5.22"],
                        [nemesisversion="v5.22"],
                        [nemesisversion="v3.09"])
                ])

  # Trump --enable-nemesis with --disable-mpi
  AS_IF([test "x$enablempi" = xno],
        [nemesisversion=no])

  AS_CASE("${nemesisversion}",
          ["v3.09"],
          [
            dnl The NEMESIS API is distributed with libmesh, so we don't have to guess
            dnl where it might be installed...
            NEMESIS_INCLUDE="-I\$(top_srcdir)/contrib/nemesis/$nemesisversion"
            AC_DEFINE(HAVE_NEMESIS_API, 1, [Flag indicating whether the library will be compiled with Nemesis support])
            AC_MSG_RESULT(<<< Configuring library with Nemesis version $nemesisversion support >>>)
          ],
          ["v5.22"],
          [
            NEMESIS_INCLUDE="-I\$(top_srcdir)/contrib/nemesis/$nemesisversion/nemesis"
            AC_DEFINE(HAVE_NEMESIS_API, 1, [Flag indicating whether the library will be compiled with Nemesis support])
            AC_MSG_RESULT(<<< Configuring library with Nemesis version $nemesisversion support >>>)
          ],
          [
            NEMESIS_INCLUDE=""
            enablenemesis=no
          ])

  AC_CONFIG_FILES([contrib/nemesis/v3.09/Makefile])
  AC_CONFIG_FILES([contrib/nemesis/v5.22/nemesis/Makefile])
{'fraction_non_alphanumeric': 0.161829579753889, 'fraction_numerical': 0.016484792198746226, 'mean_word_length': 1.7544757033248082, 'pattern_counts': {'":': 0, '<': 15, '<?xml version=': 0, '>': 15, 'https://': 0, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '5848998', 'n_tokens_mistral': 1322, 'n_tokens_neox': 1206, 'n_words': 249}